明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

プログラミング演習(11)
関数の復習と発展

中村，高橋
小林，橋本

目標

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

- 関数（メソッド）の理解を深める
- 関数の引数に配列を取る！

関数（メソッド）とは

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

- 何かの処理をしてくれる「処理機械」みたいな物



なんか
処理機

「なんか処理機」の中は
良くわからないブラックボックス
だけど使える！
具材入れたらカレーができる！

例： 関数（メソッド）

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

- 電子レンジ
  - 冷えたお弁当を入れてボタンを押すと，温まったお弁当が手に入る
- ポット
  - ボタンを押すとなんかお湯が出てくる
- 冷蔵庫
  - 食材を冷蔵庫に入れる（保存する）
- ストップウォッチ
  - ボタンを押すとラップタイプが記録されているらしい

たとえば

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

- background(r, g, b);
- ellipse(cx, cy, rx, ry);
- dist(x1, y1, x2, y2);
- pow(x, n);
- radians(theta);
- sin(radians(theta));
- atan2(mouseY - cy, mouseX - cx);
- hour(); minute(); second();

とかいろいろありましたよね？

関数ってなんで必要なの？

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

- 同じ処理を何度も書いていませんか？
- drawがやたらめったら長くなっていませんか？
- どこからどこまでがちゃんと動くか分離できていなくはないですか？
- 一度関数を作ったら使いまわせるよ！

メソッドは4種類

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



わかりにくいけれど

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

- メソッドは何らかの出力をするのでは？
  - 何かを画面に表示する
  - 何か音を鳴らす
  - 何かをファイルに出力する
  - クラス内変数の値を変更する
  - グローバル変数の値を何かに変更する

  出力してるじゃん！！
- 上記はメソッドの明示的な出力ではなく副産物みたいなもの．メソッドとしての明示的な出力はある場合とない場合がある．

メソッドの内部処理

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

- 引数として取得した値は，引数で指定された変数名を利用して処理．returnで何か返される



メソッドの内部処理

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

- 引数は増えるけれど返り値（returnされる値）は増えない



メソッドの内部処理

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

- 引数は増えるけれど返り値（returnされる値）は増えない



メソッドの内部処理

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

- 返り値だけのケース（引数がない場合）




```
if(NankaSyori() == ???){ … }
```

```
int NankaShori()
{
  グローバル変数とかを利用して処理
  return なにか返す;
}
```

メソッドの内部処理

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

- 入力も出力もないメソッド




```
NankaSyori();
```

```
void NankaShori()
{
  なんかグローバル変数を利用して処理
}
```

もう一度整理

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

何か入力して
何か出力される

何か入力されるが
何も出力されない

何も入力してないけど
何か出力される

何も入力してないし
何も出力されない

return

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

- return があるとそこで関数を抜け出す！

```
int hogehogeFunc(int nya, int piyo)
{
  // 色々な処理

  if( nya == ?? )
  {
    // 色々な処理
    return 2;
  }

  // 色々な処理
  return 3;
}
```

この if 文の中に入ったら
return 2 して終わり！

↑の if 文の判定で含まれ
なかったらここに来る！

引数：有，返り値：有

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

(Q) ある入力された数字の約数の数を求める関数をどう作るか？ また，その関数を使って数字と約数の数のペアを出力しよう

```
17989: 2
17990: 16
17991: 6
17992: 16
17993: 4
17994: 8
17995: 8
17996: 12
17997: 8
17998: 4
17999: 4
18000: 60
```

約数の数を求める関数

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

- 考え方
  - 引数は整数型の num にする
  - 約数を数える整数型の変数 count を用意
  - 整数型の変数 i （1からnumまで1ずつ増やす）を用意し，num が i で割り切れたら count を1追加する
  - 最後に count の値を返す（return count;）
  - println で数と，返って来た値を表示する

```
int getNumberOfDivisor(int num)
{
  int i=1;
  int count=0;
  while(i <= num)
  {
    if(num % i == 0)
    {
      count++;
    }
    i++;
  }
  return count;
}

void setup()
{
  for(int i = 1; i < 100000; i++)
  {
    println(i + ": " + getNumberOfDivisor(i));
  }
}
```

返り値（この値が，呼び出し元に返る）

ここに戻る

引数：有，返り値：無

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

（Q）x，y 座標を指定すると棒人間を描いてくれる関数を作成せよ！





小テスト

棒人間を描く

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

- 考え方
  - 棒人間は，顔の中心の座標（x，y）を与えると，勝手に体と手と足を描くものにする
  - 棒人間の中心の座標を（x，y）としたときのそれぞれの座標を決める
  - 返り値はなし（void）
    - 描画するだけ



棒人間を描く

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室


```
void setup()
{
  size(400, 400);
}

void drawHuman(int x, int y)
{
  ellipse(x, y, 20, 20);
  line(x, y + 10, x, y + 40);
  line(x - 20, y + 10, x + 20, y + 10);
  line(x, y + 40, x - 20, y + 60);
  line(x, y + 40, x + 20, y + 60);
}

void draw()
{
  background(255, 255, 255);
  drawHuman(mouseX, mouseY);
}
```

引数：無，返り値：有

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

(Q) 3人がランダムに手を出すじゃんけんをして，引き分けた場合は0，誰かが勝った場合は1を返す関数を作成し，1000回実施することで引き分ける確率を求めよ

- `int getJankenResult();` という関数を作ろう！

何秒目かを求める関数

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

- 考え方
  - boolean型（true/false）のstartingという変数を用意
    - starting == falseなら動かない，starting == trueなら動く
  - int型のstartTimeを用意し，マウスクリックされるとstartTime に getNow() を代入
  - int型のendTimeを用意し，マウスクリックされるとendTime に getNow() を代入
  - starting == trueの間は，drawで getNow()-startTimeの値を表示する！

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室


```
int getNow()
{
  return (hour() * 60+minute()) * 60 + second();
}

boolean start = false;
int startTime = 0;
int endTime = 0;
void setup()
{
  size(600, 300);
  textSize(120);
}

void draw()
{
  background(255, 255, 255);
  fill(0);
  if (start == true)
  {
    text((getNow() - startTime), 30, 150);
  }
  else
  {
    text((endTime - startTime), 30, 150);
  }
}
```

```
void mousePressed()
{
  if (start == false)
  {
    startTime = getNow();
    start = true;
  }
  else if (start == true)
  {
    endTime = getNow();
    start = false;
  }
}
```

getNow() 必要？

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

- 秒は `(hour() * 60 + minute()) * 60 + second()` で計算できるので getNow() はいらないのでは？
- 毎回書くのは面倒だけど，コピペしたらいいし

本当にそれで良いですか？

```
void draw()
{
  background(255, 255, 255);
  fill(0);
  if (start == true)
  {
    text(((hour() * 60 + minute()) * 60 + second() - startTime), 30, 150);
  }
  else
  {
    text((endTime - startTime), 30, 150);
  }
}
```

```
void draw()
{
  background(255, 255, 255);
  fill(0);
  if (start == true)
  {
    text((getNow() - startTime), 30, 150);
  }
  else
  {
    text((endTime - startTime), 30, 150);
  }
}
```

こちらなら
メソッドにした
方が良い！

引数：無，返り値：無

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

(Q) マウスカーソルの位置に棒人間を描いてくれる関数を作成せよ！





引数：無，返り値：無

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

- mouseX，mouseY はグローバル変数だし・・

```
void setup()
{
  size(400, 400);
}
void drawHuman2()
{
  ellipse(mouseX, mouseY, 20, 20);
  line(mouseX, mouseY + 10, mouseX, mouseY + 40);
  line(mouseX - 20, mouseY + 10, mouseX + 20, mouseY + 10);
  line(mouseX, mouseY + 40, mouseX - 20, mouseY + 60);
  line(mouseX, mouseY + 40, mouseX + 20, mouseY + 60);
}
void draw(){
  background(255, 255, 255);
  drawHuman2();
}
```

グローバル変数を使えば何もなくても良いが，改良しにくい

配列の平均を求める関数

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

(Q) 配列の平均値を求める関数を作れ

- 配列として与えられる値の平均値を求める関数を作成しよう
- 配列の要素数は気にせずに実行できるようにせよ！
- 合計値を計算して，配列の大きさで割る！

```
float getAverage(int[] array)
{
  // 合計用の変数を用意
  float total=0.0;
  for(int i=0; i<array.length; i++)
  {
    total+=array[i];
  }
  // 合計を配列の大きさで割れば平均！
  return total/array.length;
}
```

```
void setup()
{
  int[] score1 = {80, 90, 80, 55, 100, 30, 55};
  int[] score2 = {90, 50, 70, 65, 80};
  // score1の平均を求める
  println("avg1:" +getAverage(score1));
  // score2の平均を求める
  println("avg2:" +getAverage(score2));
}
```

配列の最大値を求める関数

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

(Q) 配列の平均値を求める関数を作れ

- 配列として与えられる値の平均値を求める関数を作成しよう
- 配列の要素数は気にせずに実行できるようにせよ！
- 合計値を計算して，配列の大きさで割る！

```
int getMaximum(int[] array)
{
  int max=array[0];
  for(int i=1; i<array.length; i++)
  {
    if(max<array[i])
    {
      max=array[i];
    }
  }
  // maxを返す！
  return max;
}
```

```
void setup()
{
  int[] score1 = {80, 90, 80, 55, 100, 30, 55};
  int[] score2 = {90, 50, 70, 65, 80};
  // score1の最大値を求める
  println("avg1:" +getMaximum(score1));
  // score2の最小値を求める
  println("avg2:" +getMaximum(score2));
}
```

配列を並び替える

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

(Q) 配列の値をバブルソートを利用して昇順に並び替える関数を作成せよ

```
// 配列をソートするメソッド
void bubble_sort(int[] array)
{
  for(int i=0; i<array.length; i++)
  {
    for(int j=array.length-1; j>i; j--)
    {
      if(array[i]>array[j])
      {
        int temp=array[i];
        array[i]=array[j];
        array[j]=temp;
      }
    }
  }
}
```

```
void setup()
{
  int[] score = {80, 90, 80, 55, 100, 30, 55};
  println(score);
  bubble_sort(score);
  println(score);
}
```

演習

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室

- 500x500のウインドウに，50ピクセルずつあけて九九の表を作成せよ
  - 九九の表の数字が平方数の場合は，赤色の文字で，それ以外の文字は黒色の文字で描くようにせよ
  - 平方数かどうかを判定するメソッド（引数を判定する数字，返り値をboolean型で，引数で指定した数字が平方数の場合はtrueを，平方数でない場合はfalseを返すものとせよ）も作成し，利用せよ．
  - なお，平方数の判定は単純に，入力された数字について1から順に2乗の値と一致するかどうかで判定せよ
- 座標に関する2次元配列の値を引数として，XYそれぞれの軸に水平で最小となる長方形を描画せよ